#关闭大逃杀对局
scoreboard players set IsTheDieOutGame SystemAdmin 0

#设置世界边界信息
worldborder set 6666666 0
worldborder warning distance 2
worldborder damage amount 4

#清空所有玩家触发器
scoreboard players set @a DieOutTrigger 0
#设置玩家状态为强制死亡
scoreboard players set @a SystemDeath 2

#清空DieOutInfo计分板数据
scoreboard objectives setdisplay sidebar DataKillCount
scoreboard players reset AlivePlayerCount DieOutInfo
scoreboard players reset UpdateCount DieOutInfo
scoreboard players reset TimeClock DieOutInfo


##在DieOutCenter当前刷圈中心重置有关于位置的信息
execute as @e[type=armor_stand,tag=DieOutCenter] at @s run spreadplayers ~ ~ 0 1 false @s
execute at @e[type=armor_stand,tag=DieOutCenter] run setworldspawn ~ ~1 ~
execute at @e[type=armor_stand,tag=DieOutCenter] run worldborder center ~ ~
execute at @e[type=armor_stand,tag=DieOutCenter] run spreadplayers ~ ~ 1 5 false @a
##若还未生成游戏则将玩家在原地传送，避免意外
execute unless entity @e[type=armor_stand,tag=DieOutCenter] as @a at @s run spreadplayers ~ ~ 0 1 false @s

#清空队伍
team empty DieOutTeam

#清空所有玩家关于大逃杀的tag标签
tag @a remove DieOutSelectFinish
tag @a remove DieOutWatch
tag @a remove DieOutAgree



#清空所有关于大逃杀的实体
kill @e[type=armor_stand,tag=DieOutCenter]
kill @e[type=armor_stand,tag=RunAwayCenter]
